Interferometric properties of the state extractable via an operation commuting with the interferometer

In this notebook we will present calculations involving the properties of a (0+2)-photon approximation to squeezed vacuum, and show that it can be extracted using operations commuting with a balanced interferometer.

Matrices definition

In the Heisenberg picture, one can interpret the action of passive optical elements as modifying the annihilation/creation operators with matrices. The output annihilation/creation operators are described as
$(\hat{a}'_1, \ldots, \hat{a}'_n, a'^\dagger_1, \ldots, a'^\dagger_n)^T = U (\hat{a}_1, \ldots, \hat{a}'_n, a^\dagger_1, \ldots, a^\dagger_n)^T$, where U is an unitary operator describing the passive element. In the case of two modes, the following unitaries acting on $(a_1, a_2, a^\dagger_1, a^\dagger_2)$ describe the two basic elements: *beamsplitter* (matrix **B**), and *balanced phase shift* (matrix **P**).
We are using the 4×4 matrices for easier calculations; this description also allows for easy generalization to squeezing operations. For *passive* elements, the matrices split into diagonal blocks as $U = V \oplus V^*$, where V acts on the annihilation operators and the conjugate matrix V^*acts on creation operators.

In[1]:=
$$B = \frac{\begin{pmatrix} 1 & i & 0 & 0 \\ i & 1 & 0 & 0 \\ 0 & 0 & 1 & -i \\ 0 & 0 & -i & 1 \end{pmatrix}}{\sqrt{2}}; \text{(*beamsplitter*)}$$

$$P = \begin{pmatrix} \text{Exp}[i\,\phi/2] & 0 & 0 & 0 \\ 0 & \text{Exp}[-i\,\phi/2] & 0 & 0 \\ 0 & 0 & \text{Exp}[-i\,\phi/2] & 0 \\ 0 & 0 & 0 & \text{Exp}[i\,\phi/2] \end{pmatrix}; \text{(*balanced phase shift*)}$$

We will manipulate expressions involving comp
lex products of annihilation and creation operators with the internal **Dot** function (used in Wolfram language to denote the standard matrix product), and they are not automatically handled with Mathematica. The following function defines rules simplifying such expressions: removal of numerical factors outside of the dot product, commutation rules, and mode index ordering.

In[3]:=
```
simprules[order_ : {2, 1}] :=
  {Dot[ops1___, prefactor_ op_, ops2___] /; (MemberQ[{id, a1, A1, a2, A2}, op]) :>
    prefactor Dot[ops1, op, ops2]
   (*move non-operator prefactors outside. a.(prefactor b)=prefactor a.b*),
   Dot[ops1___, prefactor_ op_, ops2___] /; Head[op] == Dot :> prefactor Dot[ops1, op, ops2]
   (*collapse stacked dot products, a.(prefactor b.c).d=prefactor a.b.c.d*),
   Dot[ops1_, id, ops2_] :> Dot[ops1, ops2](*collapse identity operators, ×3*),
   Dot[id, ops1_, ops2_] :> Dot[ops1, ops2],
   Dot[ops1_, ops2_, id] :> Dot[ops1, ops2],

   Dot[ops1___, ai_, Aj_, ops2___] /; i == j :> Dot[ops1, Aj, ai, ops2] + Dot[ops1, id, id, ops2]
   (*move creation operators to the left,
   adding identity due to commutation rules*),
   Dot[ops1___, ai_, Aj_, ops2___] /; i ≠ j :> Dot[ops1, Aj, ai, ops2]
   (*move creation ops to the left - they commute for different modes*),
   Dot[ops1___, ai_, aj_, ops2___] /; order[[i]] > order[[j]] :> Dot[ops1, aj, ai, ops2],
   (*enforce mode index ordering for easier calculations, ×2*)
   Dot[ops1___, Ai_, Aj_, ops2___] /; order[[i]] < order[[j]] :> Dot[ops1, Aj, Ai, ops2],

   pref_Dot[ai_, id] :> pref Dot[id, ai](*move identity operators to the center,
   they have to be preserved for easier symbolic calculations*),
   pref_Dot[id, Ai_] :> pref Dot[Ai, id]
  };
```

If the total expression is presented as sum of terms in the normal ordering (each term having form $a^\dagger_{i_1} \ldots a_{i_k}$), only the terms containing *identity operators alone* contribute to the vacuum expectation value: $\langle 0 |\ a^\dagger_{i_1} \ldots a_{i_k}\ | 0 \rangle = 0$, but $\langle 0 | 0 \rangle = 1$. The following function defines the rules to simplify the terms (for specific mode **indices** – later it will be useful to assume that e.g. mode 1 is in vacuum, but mode 2 is in a superposition of finitely many Fock states), leaving only products of identity operators.

In[4]:=
```
expvacuumrules[indices_, order_ : {2, 1}] :=
  simprules[order]~Join~{pref_Dot[ops1__, ai_] /; MemberQ[indices, i] :> 0,
    pref_Dot[Ai_, ops1__] /; MemberQ[indices, i] :> 0,
    pref_id.id.id :> pref Dot[id, id]};
```

So, the final expression for vacuum expectation value is calculated by the following rule, turning identity operators into ones.

In[5]:=
```
finalizerules = {pref_id.id :> pref 1};
```

Unsqueezed light

First, let us calculate the phase determination error of an interferometer having coherent state $|\gamma\rangle$ as one of its inputs, while the second mode is in vacuum. For consistency, we want to use only the vacuum expectation values, and thus the first part is to perform the displacement operation $D(\gamma)$, turning a_1 into $a_1 + \gamma$.

In[6]:=
```
{b1, b2, β1, β2} = {a1 + γ id, a2, A1 + γ* id, A2};
```

Then, the final modes at the interferometer output are modified by the matrices described above. Expectation value of the photon number difference $a_2'^\dagger a_2' - a_1'^\dagger a_1$ at the interferometer output will be calculated from the following expression **exprC**:

In[7]:=
```
{b1, b2, β1, β2} = Expand /@ FullSimplify[B*.P.B.{b1, b2, β1, β2}];
exprC = Expand[(Distribute[(β2).(b2)] - Distribute[(β1).(b1)]) //. simprules[]]
```

Out[8]=
$$-\gamma\,\mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{id.id} - \mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{id.a_1} - \gamma\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{A_1.id} - \mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{A_1.a_1} + \mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{A_2.a_2} + 2\,\mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right] \mathrm{id.a_2}\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\mathrm{Cos}\left[\frac{\phi}{2}\right] \mathrm{A_1.a_2}\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\gamma\,\mathrm{Cos}\left[\frac{\phi}{2}\right] \mathrm{A_2.id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\mathrm{Cos}\left[\frac{\phi}{2}\right] \mathrm{A_2.a_1}\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + \gamma\,\mathrm{Conjugate}[\gamma]\,\mathrm{id.id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + \mathrm{Conjugate}[\gamma]\,\mathrm{id.a_1}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + \gamma\,\mathrm{A_1.id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + \mathrm{A_1.a_1}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 - \mathrm{A_2.a_2}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2$$

... while the variance of photon number difference depends on the $\left(a_2'^\dagger a_2' - a_1'^\dagger a_1\right)^2$ (too long to print out):

In[9]:=
```
expr2C = Expand[Distribute[exprC.exprC] //. simprules[]];
```

The expectation values of the photon number, as well as its second power, and the resulting variance are calculated here to be used later:

In[10]:=
```
firstmomCOH = exprC //. expvacuumrules[{1, 2}] /. finalizerules // FullSimplify;

secmomCOH = expr2C //. expvacuumrules[{1, 2}] /. finalizerules;
varCOH = secmomCOH - firstmomCOH^2 // Expand // FullSimplify // Expand;
Print["⟨N2-N1⟩: ", firstmomCOH]
Print["variance of N2-N1: ", varCOH]
```

```
⟨N2-N1⟩: -γ Conjugate[γ] Cos[ϕ]
variance of N2-N1: Abs[γ]^2
```

We recover the standard result that the variance is $|\gamma|^2$ for the input coherent state $|\gamma\rangle$.

Squeezed light

If the originally vacuum input mode 2 is *squeezed*, the operators A_2, a_2 are modified by the following matrix:

In[15]:=
```
S = {{1, 0, 0, 0}, {0, Cosh[r], 0, -Sinh[r]}, {0, 0, 1, 0}, {0, -Sinh[r], 0, Cosh[r]}};
```

$$S = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \text{Cosh}[r] & 0 & -\text{Sinh}[r] \\ 0 & 0 & 1 & 0 \\ 0 & -\text{Sinh}[r] & 0 & \text{Cosh}[r] \end{pmatrix};$$

The first mode is, as before, assumed to be coherent $|\gamma\rangle$. We can calculate the expectation values using the same procedure:

In[16]:=
```
{b1, b2, β1, β2} = Expand /@ FullSimplify[S.{a1, a2, A1, A2}];
{b1, b2, β1, β2} = {b1, b2, β1, β2} + {id γ, 0, id γ*, 0};
{b1, b2, β1, β2} = Expand /@ FullSimplify[B†.P.B.{b1, b2, β1, β2}];
exprS = Expand[(Distribute[(β2).(b2)] - Distribute[(β1).(b1)]) //. simprules[]];
expr2S = Expand[Distribute[exprS.exprS] //. simprules[]];

firstmomSQ = exprS //. expvacuumrules[{1, 2}] /. finalizerules;
secmomSQ = expr2S //. expvacuumrules[{1, 2}] /. finalizerules;
varSQ = secmomSQ - firstmomSQ^2 // Expand // FullSimplify // Expand;
Print["⟨N2-N1⟩: ", firstmomSQ]
Print["variance of N2-N1: ", varSQ]
```

$\langle N_2\text{-}N_1\rangle$: $-\gamma\,\text{Conjugate}[\gamma]\,\text{Cos}\left[\frac{\phi}{2}\right]^2 + \gamma\,\text{Conjugate}[\gamma]\,\text{Sin}\left[\frac{\phi}{2}\right]^2 + \text{Cos}\left[\frac{\phi}{2}\right]^2 \text{Sinh}[r]^2 - \text{Sin}\left[\frac{\phi}{2}\right]^2 \text{Sinh}[r]^2$

variance of $N_2\text{-}N_1$: $\frac{1}{2}\gamma\,\text{Conjugate}[\gamma] + \frac{1}{2}\gamma\,\text{Conjugate}[\gamma]\,\text{Cosh}[2\,r] - \gamma^2\,\text{Cosh}[r]\,\text{Sin}[\phi]^2\,\text{Sinh}[r] - \text{Conjugate}[\gamma]^2\,\text{Cosh}[r]\,\text{Sin}[\phi]^2\,\text{Sinh}[r] + \frac{3\,\text{Sinh}[r]^2}{4} - \frac{1}{4}\text{Cos}[2\,\phi]\,\text{Sinh}[r]^2 - \gamma\,\text{Conjugate}[\gamma]\,\text{Cos}[2\,\phi]\,\text{Sinh}[r]^2 + \frac{1}{2}\text{Cos}[\phi]^2\,\text{Sinh}[r]\,\text{Sinh}[3\,r]$

Quite complicated, but they are closed form expressions. As we can observe in the following plots of $\langle N_2 - N_1\rangle$, maximum sensitivity (measured by the derivative with respect to interferometric phase change ϕ) appears around $\phi = \frac{\pi}{2}$.

In[26]:=
```
Block[{γ = 10, r = 2},
 Plot[{firstmomCOH, firstmomSQ}, {ϕ, 0, π},
  PlotLabels → {"coherent |γ=" <> ToString[γ] <> "⟩⊗ vacuum",
    "coherent |γ=" <> ToString[γ] <> "⟩⊗ squeezed vacuum |r=" <> ToString@r <> ">"},
  ImageSize → Large, AxesLabel → {"ϕ", "⟨N2-N1⟩"}]
]
```

Out[26]=



Error of phase determination around $\phi = \frac{\pi}{2}$ is calculated using the standard error propagation formula: it is proportional to the square root of variance, and inverse proportional to the derivative of $\langle N_2 - N_1 \rangle$ with respect to ϕ. Squeezing improves the phase determination uncertainty significantly:

In[27]:=
```
Block[{γ = 10, θ = π / 2},
 errCOH = Sqrt[varCOH] / D[firstmomCOH, ϕ] /. ϕ → θ;
 errSQ = Sqrt[varSQ] / D[firstmomSQ, ϕ] /. ϕ → θ;
 improvement = errSQ / errCOH;
 Plot[{errCOH, errSQ}, {r, -3, 3}, PlotRange → {All, {0, 1}},
  AxesLabel → {"squeezing parameter r", "phase estimation inaccuracy"},
  PlotLabels → {"coherent |γ=" <> ToString[γ] <> "⟩⊗ vacuum",
    "coherent |γ=" <> ToString[γ] <> "⟩⊗ squeezed vacuum |r>"}, ImageSize → Large]
]
```

Out[27]=



Zero+One approximation to squeezed light

In the article, we show that a finite photon number approximation to the squeezed vacuum represented as a superposition of two Fock states ($|\psi\rangle = \cos\tau\,|0\rangle + \sin\tau\,|2\rangle$) also improves accuracy to some degree, and can be probabilistically prepared by a *device commuting with the interferometer given input* $|\gamma\rangle\otimes|\text{vacuum}\rangle$. This means that the device can be placed before the interferometer (and used to prepare the input state $|\gamma\rangle\otimes|\psi\rangle$), or transform the output state of the interferometer (product of coherent states) *into a result of feeding the interferometer with* $|\gamma\rangle\otimes|\psi\rangle$. In short: postprocessing is equivalent to preparation, given certain mathematical criteria. In this section we show that the input state $|\gamma\rangle\otimes|\psi\rangle$ offers a reduction in phase estimation inaccuracy (compared to $|\gamma\rangle\otimes|\text{vacuum}\rangle$) of factor $\sqrt{3-\sqrt{6}}$ (asymptotically). First, the calculations are done as previously:

In[28]:=
```
{b1, b2, β1, β2} = Expand /@ FullSimplify[{a1, a2, A1, A2}, Assumptions → {ϕ > 0, r > 0}];
{b1, b2, β1, β2} = {b1, b2, β1, β2} + {γ id, 0, γ* id, 0};
{b1, b2, β1, β2} =
  Expand /@ FullSimplify[B†.P.B.{b1, b2, β1, β2}, Assumptions → {ϕ > 0, r > 0}];
% // FullSimplify // MatrixForm;
expr01 = Expand[(Distribute[(β2).(b2)] - Distribute[(β1).(b1)]) //. simprules[]];
expr201 = Expand[Distribute[expr01.expr01] //. simprules[]];
```

Remember that **expr01** is a polynomial in the annihilation and creation operators of the input modes (and **expr201** is its square):

In[34]:=
```
expr01
```

Out[34]=

$$-\gamma\,\mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{id}.\mathrm{id} - \mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 \mathrm{id}.a_1 - \gamma\,\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 A_1.\mathrm{id} - \mathrm{Cos}\left[\frac{\phi}{2}\right]^2 A_1.a_1 + \mathrm{Cos}\left[\frac{\phi}{2}\right]^2 A_2.a_2 + 2\,\mathrm{Conjugate}[\gamma]\,\mathrm{Cos}\left[\frac{\phi}{2}\right] \mathrm{id}.a_2\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\mathrm{Cos}\left[\frac{\phi}{2}\right] A_1.a_2\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\gamma\,\mathrm{Cos}\left[\frac{\phi}{2}\right] A_2.\mathrm{id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + 2\,\mathrm{Cos}\left[\frac{\phi}{2}\right] A_2.a_1\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + \gamma\,\mathrm{Conjugate}[\gamma]\,\mathrm{id}.\mathrm{id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + \mathrm{Conjugate}[\gamma]\,\mathrm{id}.a_1\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + \gamma\,A_1.\mathrm{id}\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 + A_1.a_1\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2 - A_2.a_2\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2$$

We now have to assume that the input mode 1 is in vacuum (which is then displaced by $D(\gamma)$ and fed into the interferometer), and the monomials ending with a_1 (or beginning with $a_1^\dagger$) do not contribute to the expectation values, while we replace the operators a_2 and $a_2^\dagger$ with finite-dimensional matrices (which leads to strict result, since the state has finite maximal photon number), and calculate the expectation values of with the input state $|\psi\rangle = \cos\tau\,|0\rangle + \sin\tau\,|2\rangle$.

First, let’s define the state and the matrices:

In[35]:=
```
maxdim = 8;
ANN[dim_] := DiagonalMatrix[Table[√k , {k, dim - 1}], 1];
vec01 = PadRight[{Cos[τ], 0, -Sin[τ]}, maxdim];
covec01 = vec01† // ComplexExpand;
Print["truncated dimension = ", maxdim];
Print["2-photon approximation = ", vec01 // MatrixForm];
Print["Truncated annihilation operator = ", ANN[maxdim] // MatrixForm];
```

truncated dimension = 8

2-photon approximation = $\begin{pmatrix} \text{Cos}[\tau] \\ 0 \\ -\text{Sin}[\tau] \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$

Truncated annihilation operator = $\begin{pmatrix} 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \sqrt{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \sqrt{3} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \sqrt{5} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \sqrt{6} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \sqrt{7} \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

The expectation values are calculated in the following step.

In[42]:=
```
finalizetomatrices[idx_, dim_] := {a_idx → ANN[dim], A_idx → ANN[dim]†, id → IdentityMatrix[dim]}

firstmom01OP = expr01 //. expvacuumrules[{1}] /. finalizetomatrices[2, maxdim];
secmom01OP = expr201 //. expvacuumrules[{1}] /. finalizetomatrices[2, maxdim];

firstmom01 = covec01.firstmom01OP.vec01;
secmom01 = covec01.secmom01OP.vec01;
var01 = secmom01 - firstmom01^2 // Expand // FullSimplify // Expand
```

Out[47]=

$$2\,\gamma\,\text{Conjugate}[\gamma] - \gamma\,\text{Conjugate}[\gamma]\,\text{Cos}[2\,\tau] + \frac{3\,\text{Sin}[\tau]^2}{2} - \frac{1}{2}\,\text{Cos}[2\,\phi]\,\text{Sin}[\tau]^2 - 2\,\gamma\,\text{Conjugate}[\gamma]\,\text{Cos}[2\,\phi]\,\text{Sin}[\tau]^2 + \text{Cos}[\phi]^2\,\text{Sin}[\tau]\,\text{Sin}[3\,\tau] - \sqrt{2}\,\gamma^2\,\text{Cos}[\tau]\,\text{Sin}[\tau]\,\text{Sin}[\phi]^2 - \sqrt{2}\,\text{Conjugate}[\gamma]^2\,\text{Cos}[\tau]\,\text{Sin}[\tau]\,\text{Sin}[\phi]^2$$

As before, the maximal accuracy of phase determination is attained around $\phi = \frac{\pi}{2}$:

In[48]:=
```
Block[{γ = 5, τ = π / 2},
 Plot[{firstmomCOH, firstmom01}, {ϕ, 0, π},
  PlotLabels → {"coherent |γ=" <> ToString[γ] <> "⟩⊗ vacuum",
    "coherent |γ=" <> ToString[γ] <> "⟩⊗ |ψ (τ=π/2)>"},
  ImageSize → Large, AxesLabel → {"ϕ", "⟨N2-N1⟩"}]
]
```

Out[48]=



And there is a modest improvement in phase determination inaccuracy around this point:

In[49]:=
```
Block[{γ = 100, θ = π/2},
  errCOH = Sqrt[varCOH] / D[firstmomCOH, ϕ] /. ϕ → θ;
  err01 = Sqrt[var01] / D[firstmom01, ϕ] /. ϕ → θ;
  Plot[{errCOH, err01}, {τ, 0, π/2}, PlotRange → {All, {0, 2 errCOH}},
   AxesLabel → {"state parameter τ", "phase estimation inaccuracy"},
   PlotLabels → {"coherent |γ=" <> ToString[γ] <> "⟩⊗ vacuum",
     "coherent |γ=" <> ToString[γ] <> "⟩⊗ |ψ(τ)⟩"}, ImageSize → Large]
]
```

Out[49]=



The asymptotic improvement of accuracy can be determined analytically. Let's assume for simplicity that the input coherent state has real positive displacement γ, and calculate the **improvement** as a function of γ and the parameterization angle τ:

In[50]:=
```
errCOH = Sqrt[varCOH] / D[firstmomCOH, ϕ];
err01 = Sqrt[var01 // ComplexExpand] / D[firstmom01, ϕ];
improvement = FullSimplify[err01 / errCOH /. ϕ → π/2, Assumptions → {γ > 0}];
Print["Improvement = ", improvement];
```

$$\text{Improvement} = \frac{\gamma\sqrt{1-\left(1+2\gamma^2\right)\text{Cos}[2\tau]+\gamma^2\left(3-\sqrt{2}\ \text{Sin}[2\tau]\right)}}{-1+\gamma^2+\text{Cos}[2\tau]}$$

The minimum (and extremum, removed manually) can be determined by postulating that the derivative with respect to τ vanishes. The optimal τ can be determined for arbitrary γ, although it is a rather involved function of it. With increasing γ it quickly stabilizes, so let's determine its asymptotic value instead. For this, let's calculate the derivative of **improvement** (squared, for easier calculations) and expand it around $\gamma \to \infty$, and then find the solutions:

```
In[54]:= dd = D[improvement^2, τ] // FullSimplify
dd = FullSimplify[Series[dd, {γ, ∞, 1}] // Normal, Assumptions → {γ > 0}];
sols = (Solve[dd == 0, τ] // Normal) /. {c1 → 0};
```

Out[54]=

$$\frac{1}{\left(-1+\gamma^2+\mathrm{Cos}[2\,\tau]\right)^3}\,\gamma^2\left(-3\sqrt{2}\,\gamma^2-2\sqrt{2}\,\gamma^2\left(-1+\gamma^2\right)\mathrm{Cos}[2\,\tau]+\sqrt{2}\,\gamma^2\,\mathrm{Cos}[4\,\tau]+2\left(1+5\,\gamma^2+2\,\gamma^4\right)\mathrm{Sin}[2\,\tau]-\left(1+2\,\gamma^2\right)\mathrm{Sin}[4\,\tau]\right)$$

The first solution corresponds to the minimum. It is:

```
In[57]:= Print["Asymptotically optimal τ = ", τ /. sols[[1]]];
Print["For this τ, improvement = ", improvement /. sols[[1]] // FullSimplify]
Print["Asymptotic improvement = ", Limit[improvement /. sols[[1]], γ → ∞] // FullSimplify]
Plot[{improvement /. sols[[1]], Sqrt[3 - Sqrt[6]]}, {γ, 1, 10},
 PlotRange → {All, {0, 1}}, PlotStyle → {Automatic, {Dotted, Black}},
 AxesLabel → {"Input coherent state amplitude γ", "phase estimation improvement"},
 ImageSize → Large]
```

Asymptotically optimal $\tau = \frac{1}{2}\,\mathrm{ArcTan}\left[\frac{1}{\sqrt{2}}\right]$

For this τ, improvement $= \frac{\sqrt{9-3\sqrt{6}}\;\gamma\sqrt{1+3\,\gamma^2}}{-3+\sqrt{6}+3\,\gamma^2}$

Asymptotic improvement $= \sqrt{3-\sqrt{6}}$

Out[60]=



Characteristic functions

Such a state offers metrological advantage, and here we show that it can be produced from bimodal input of coherent and vacuum states. First, let’s calculate its characteristic function with respect to

the $U(1)$ group corresponding to the interferometer action. The $U(1)$ parameter is exactly the angle ϕ of differential phase shift inside the interferometer.
The $U(1)$ group is a subgroup of the overall $SU(2)$ structure, related to arbitrary mode mixing operations. Let us notice that the matrix describing the interferometer action on a_1, a_2 operators is indeed a standard 2D rotation, which can be thought of as $U(1)$:

In[61]:=
```
(B†.P.B)[[{1, 2}, {1, 2}]] // FullSimplify // MatrixForm
```

Out[61]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}\left[\frac{\phi}{2}\right] & -\mathrm{Sin}\left[\frac{\phi}{2}\right] \\ \mathrm{Sin}\left[\frac{\phi}{2}\right] & \mathrm{Cos}\left[\frac{\phi}{2}\right] \end{pmatrix}$$

Comparing it to the most general form of $SU(2)$ matrix, $\begin{pmatrix} u & -v^* \\ v & u^* \end{pmatrix}$, we see that $u = \cos\frac{\theta}{2}$, $v = \sin\frac{\theta}{2}$.
With $z = u + i\,v = \exp\left(i\,\theta/2\right)$ (and the constraint that $|z| = 1$), one can write the matrix as $\frac{1}{2z}\begin{pmatrix} 1+z^2 & -i(-1+z^2) \\ i(-1+z^2) & 1+z^2 \end{pmatrix}$, and this can be directly used in the calculation of the characteristic functions related to the interferometer action.

The initial state is $|\gamma\rangle \otimes |\psi\rangle$, where $\left|\psi\right\rangle = \cos\tau\left|0\right\rangle + \sin\tau\left|2\right\rangle$ and $\left|\gamma\right\rangle = e^{-|\alpha|^2}\sum_n \frac{\alpha^n}{\sqrt{n!}}\left|n\right\rangle$ is a coherent state, and in the product basis the total state can be written as

$$\left|\gamma\right\rangle \otimes \left|\psi\right\rangle = e^{-|\alpha|^2}\left(|0,0\rangle\cos\tau + \alpha\left|1,0\right\rangle\cos\tau + \sum_{n=2}\left[\frac{\alpha^n}{\sqrt{n!}}\left|n,0\right\rangle\cos\tau - \frac{\alpha^{n-2}}{\sqrt{(n-2)!}}\left|n-2,2\right\rangle\sin\tau\right]\right).$$

Thus, for characteristic function we need only the matrix elements of the $SU(2)$ unitary representation between $|n,0\rangle$ and $|n-2,2\rangle$ states. They can be found explicitly, and the following code defines the matrix elements (as a 2×2 matrix) in the function **RedSU2**.

```
In[62]:= statetopoly[j_, ψ_, {z1_, z2_}] :=
  Sum[(-1)^(j - m) Sqrt[Binomial[2 j, j - m]] ψ[[(j - m) + 1]] z1^(j + m) z2^(j - m), {m, -j, j}]
 (* Majorana polynomial for a given state ψ*)
 transformpoly[poly_, {z1_, z2_}] := poly /. Thread[{z1, z2} → ({z1, z2}.( u  -v
                                                                           V   U ))]

 getcoeff[j_, poly_, {z1_, z2_}] := Module[{coeff},
   coeff = Reverse@CoefficientList[poly, {z1, z2}, {2 j + 1, 2 j + 1}];
   coeff = Diagonal[coeff];
   coeff = coeff * Table[(-1)^(i + 1) / Sqrt[Binomial[2 * j, i - 1]], {i, 2 j + 1}]]
 (*state for a given polynomial*)

 SU2Unitary[j_] := Module[{basisstates, basispolys, z},
   basisstates = IdentityMatrix[2 j + 1];
   basispolys = transformpoly[statetopoly[j, #, {z1, z2}], {z1, z2}] & /@ basisstates;
   getcoeff[j, #, {z1, z2}] & /@ basispolys](*Hilbert state picture of
  the transformation defined in g: how are basis states transformed?*)
 SU2Unitary[jmin_, jmax_] :=
  DiagonalMatrix[Hold /@ Table[SU2Unitary[j], {j, jmin, jmax, 1 / 2}]] // ReleaseHold //
   ArrayFlatten
 SUB1Unitary[j_] :=
  SU2Unitary[j] /. {u → (z + 1 / z) / 2, v → ⅈ (z - 1 / z) / 2, U → (z + 1 / z) / 2, V → -ⅈ (1 / z - z) / 2} // Expand
```

$$\texttt{RedSU2[j\_] :=} \begin{pmatrix} u^{2j} & \sqrt{\texttt{Binomial}[2j,2]}\, u^{2j-2} V^2 \\ \sqrt{\texttt{Binomial}[2j,2]}\, u^{2j-2} v^2 & u^{2j-4}\left(u^2 U^2 - (4j-4)\, u\, U\, v\, V + (-1+j)(-3+2j)\, v^2 V^2\right) \end{pmatrix}$$

Characteristic function is parameterized by a complex variable z, as mentioned before. The characteristic function can be calculated explicitly by the following code block. Here, for simplification we reparameterize $\cos\tau \mapsto \sqrt{1-r}$, $\sin\tau \mapsto \sqrt{r}$, perform the summation, and switch back to τ variables at the end; the characteristic function $\chi(z)$ is exactly the overlap between the output of the interferometer and its input state $|\gamma\rangle \otimes |\psi\rangle$, for $z = \exp(i\,\theta/2)$.

In[69]:=
```
Block[{j, u = (z + 1/z)/2, U = (z + 1/z)/2, v = (z - 1/z)/(2 ⅈ), V = (z - 1/z)/(2 ⅈ), n}, j = n/2;
  vec = Exp[-Abs[γ]^2/2] {γ^n Sqrt[1 - r]/Sqrt[n!], -γ^(n-2)/Sqrt[(n - 2)!] Sqrt[r]};
  covec = Exp[-Abs[γ]^2/2] {Conjugate[γ]^n Sqrt[1 - r]/Sqrt[n!], -Conjugate[γ]^(n-2)/Sqrt[(n - 2)!] Sqrt[r]};
  ex0 = covec.RedSU2[j].vec // Expand;
  ex1 = ex0;
  ex2 = Simplify[ex1, Assumptions → {n ≥ 2, r > 0, r < 1}] // Expand;
  ex2 = ex2 /. Sqrt[-((-1 + n) n (-1 + r) r (-2 + n)! n!)] :→ n! Sqrt[r (1 - r)];
  ex2 = ex2 // Simplify;
  χ = Sum[ex2, {n, 2, ∞}] + (ⅇ^(-γ Conjugate[γ]) (1 - r) (2 z + (1 + z^2) γ Conjugate[γ]))/(2 z);
  χ = χ /. {r → 1 - Cos[τ]^2} // Simplify;
]
```

It is a rather complex expression involving the defining variables γ (coherent state amplitude), τ (squeezed state approximation parameter), and z (describing the interferometer phase difference angle). If we plug in the calculated (in the previous section) optimal angle we obtain:

In[70]:=
```
χopt = χ /. {τ → 1/2 ArcTan[1/Sqrt[2]]} // TrigToExp // Expand // Simplify
```

Out[70]=

$$\frac{1}{64\sqrt{6}\,z^4}\,e^{-\mathrm{Abs}[\gamma]^2-\gamma\,\mathrm{Conjugate}[\gamma]}\Bigg(8z^2\Bigg(4\left(2+\sqrt{6}\right)e^{\mathrm{Abs}[\gamma]^2}z^2-4\left(2+\sqrt{6}\right)e^{\gamma\,\mathrm{Conjugate}[\gamma]}z^2+ e^{\frac{(1+z)^2\,\gamma\,\mathrm{Conjugate}[\gamma]}{2z}}\left(-2+\sqrt{6}+\gamma^2+z^2\left(4+6\sqrt{6}-2\gamma^2\right)+z^4\left(-2+\sqrt{6}+\gamma^2\right)\right)\Bigg)+8z\left(1+z^2\right)\left(2\left(2+\sqrt{6}\right)e^{\mathrm{Abs}[\gamma]^2}z^2-2\left(2+\sqrt{6}\right)e^{\gamma\,\mathrm{Conjugate}[\gamma]}z^2+\left(-2+\sqrt{6}\right)e^{\frac{(1+z)^2\,\gamma\,\mathrm{Conjugate}[\gamma]}{2z}}\left(-1+z^2\right)^2\right)\gamma\,\mathrm{Conjugate}[\gamma]+ e^{\frac{(1+z)^2\,\gamma\,\mathrm{Conjugate}[\gamma]}{2z}}\left(-1+z^2\right)^2\left(\left(-2+\sqrt{6}\right)\gamma^2+\left(-2+\sqrt{6}\right)z^4\gamma^2+z^2\left(8-2\left(-2+\sqrt{6}\right)\gamma^2\right)\right)\mathrm{Conjugate}[\gamma]^2\Bigg)$$

For $\tau = 0$, the characteristic function corresponds to the interferometer input being $|\gamma\rangle \otimes |0\rangle$ – coherent in the first port, vacuum in the second. As it turns out, the form of the characteristic function of this state is a (rescaled) generating function of the modified Bessel functions $I_k(a)$ (see Appendix B of [1], Eq. B.51), and can be decomposed into a sum $\chi = \sum_{k \in \mathbb{Z}} z^k e^{-\gamma\gamma^*} I_k(\gamma\gamma^*)$:

In[124]:=

```
χcoh = χ /. τ → 0 // FullSimplify // PowerExpand
BesselIGeneratingFn[a_, z_] := Exp[a (z + 1 / z) / 2]
Print["Is χ_coherent equal to e^(-γ γ*) ∑_{k=-∞}^∞ z^k I_k(γγ*)? :",
 χcoh == Exp[-γ γ*] BesselIGeneratingFn[γ γ*, z] // FullSimplify]
Ck = e^(-γ Conjugate[γ]) BesselI[k, γ Conjugate[γ]];
```

Out[124]=

$$e^{\frac{(-1+z)^2\,\gamma\,\mathrm{Conjugate}[\gamma]}{2\,z}}$$

Is χ_{coherent} equal to $e^{-\gamma\,\gamma^*}\sum_{k=-\infty}^{\infty} z^k I_k(\gamma\gamma^*)$? :True

The characteristic function **χopt** of the optimal state is the one of the coherent state **χcoh** multiplied by a expression involving powers of z from z^{-4} to z^4:

In[75]:=

```
charfnprod = Collect[((χopt / (χcoh z^-4) // FullSimplify) z^-4 // Expand), z]
```

Out[75]=

$$\frac{3}{4}+\frac{1}{2\sqrt{6}}-\frac{\gamma^2}{4\sqrt{6}}+\frac{3\,\mathrm{Abs}[\gamma]^4}{32}-\frac{1}{16}\sqrt{\frac{3}{2}}\,\mathrm{Abs}[\gamma]^4+\frac{\frac{\mathrm{Abs}[\gamma]^4}{64}-\frac{\mathrm{Abs}[\gamma]^4}{32\sqrt{6}}}{z^4}+$$
$$z^4\left(\frac{\mathrm{Abs}[\gamma]^4}{64}-\frac{\mathrm{Abs}[\gamma]^4}{32\sqrt{6}}\right)-\frac{\mathrm{Conjugate}[\gamma]^2}{4\sqrt{6}}+\frac{\frac{1}{8}\,\gamma\,\mathrm{Conjugate}[\gamma]-\frac{\gamma\,\mathrm{Conjugate}[\gamma]}{4\sqrt{6}}}{z^3}+$$
$$z^3\left(\frac{1}{8}\,\gamma\,\mathrm{Conjugate}[\gamma]-\frac{\gamma\,\mathrm{Conjugate}[\gamma]}{4\sqrt{6}}\right)+\frac{-\frac{1}{8}\,\gamma\,\mathrm{Conjugate}[\gamma]+\frac{\gamma\,\mathrm{Conjugate}[\gamma]}{4\sqrt{6}}}{z}+$$
$$z\left(-\frac{1}{8}\,\gamma\,\mathrm{Conjugate}[\gamma]+\frac{\gamma\,\mathrm{Conjugate}[\gamma]}{4\sqrt{6}}\right)+\frac{\frac{1}{8}-\frac{1}{4\sqrt{6}}+\frac{\gamma^2}{8\sqrt{6}}-\frac{\mathrm{Abs}[\gamma]^4}{16}+\frac{\mathrm{Abs}[\gamma]^4}{8\sqrt{6}}+\frac{\mathrm{Conjugate}[\gamma]^2}{8\sqrt{6}}}{z^2}+$$
$$z^2\left(\frac{1}{8}-\frac{1}{4\sqrt{6}}+\frac{\gamma^2}{8\sqrt{6}}-\frac{\mathrm{Abs}[\gamma]^4}{16}+\frac{\mathrm{Abs}[\gamma]^4}{8\sqrt{6}}+\frac{\mathrm{Conjugate}[\gamma]^2}{8\sqrt{6}}\right)$$

Thus, the characteristic function **χopt** can be written as $\chi_{\text{opt}}=\sum_{k=-\infty}^{\infty} z^k P_k$, where P_k is convolution of the coherent state coefficients $C_k=e^{-\gamma\gamma^*}\,I_k(\gamma\gamma^*)$ and the following zero-centered list:

```
In[76]:= convcoeffs = FullSimplify[CoefficientList[charfnprod z^4 , z]];
convcoeffs // MatrixForm
```

Out[77]//MatrixForm=

$$\begin{pmatrix}
-\frac{1}{192}\left(-3+\sqrt{6}\right)\text{Abs}[\gamma]^4 \\
-\frac{1}{24}\left(-3+\sqrt{6}\right)\gamma\,\text{Conjugate}[\gamma] \\
\frac{1}{48}\left(6-2\sqrt{6}+\sqrt{6}\,\gamma^2+\left(-3+\sqrt{6}\right)\text{Abs}[\gamma]^4+\sqrt{6}\,\text{Conjugate}[\gamma]^2\right) \\
\frac{1}{24}\left(-3+\sqrt{6}\right)\gamma\,\text{Conjugate}[\gamma] \\
\frac{1}{96}\left(72+8\sqrt{6}-4\sqrt{6}\,\gamma^2-3\left(-3+\sqrt{6}\right)\text{Abs}[\gamma]^4-4\sqrt{6}\,\text{Conjugate}[\gamma]^2\right) \\
\frac{1}{24}\left(-3+\sqrt{6}\right)\gamma\,\text{Conjugate}[\gamma] \\
\frac{1}{48}\left(6-2\sqrt{6}+\sqrt{6}\,\gamma^2+\left(-3+\sqrt{6}\right)\text{Abs}[\gamma]^4+\sqrt{6}\,\text{Conjugate}[\gamma]^2\right) \\
-\frac{1}{24}\left(-3+\sqrt{6}\right)\gamma\,\text{Conjugate}[\gamma] \\
-\frac{1}{192}\left(-3+\sqrt{6}\right)\text{Abs}[\gamma]^4
\end{pmatrix}$$

Let' s assume that the coherent light amplitude is real an positive; then $\gamma = \sqrt{a}$ and the tails of the resulting P_k determine feasibility of the state extraction. The state can be extracted if $C_k = p\, P_k + (1-p)\, D_k$, with $0 \le p \le 1$ and $D_k \ge 0$: the extraction succeeds with probability p, and with probability $(1-p)$ a state corresponding to the probability distribution D_k is prepared. This is possible if the tails of P_k (variable **Pk**) fall off at least as quickly as C_k. Here we calculate the the P_k coefficients.

```
In[78]:= convexpr = Sum[FullSimplify[convcoeffs[[i]] /. γ → √a , Assumptions → {a > 0}]
    DiscreteDelta[k + i - 5], {i, Length@convcoeffs}];
Cks = FullSimplify[Ck /. γ → √a , Assumptions → {a > 0}];
Pk = DiscreteConvolve[Cks, convexpr, k, l] /. l → k;
```

Asymptotically (for $k \to \infty$), the Bessel function $I_k(a)$ has the expansion of $I_k(a) \approx \frac{(a/2)^k}{k!}$ (Appendix B of [1], Eq. B.47). Thus, the ratio of tails can be calculated: $\frac{P_k\left((1-\epsilon)\, a\right)}{C_k(a)} \to 0$ with $k \to \infty$ if $\epsilon > 0$. Therefore, a state $|\gamma'\rangle \otimes |\psi\rangle$ can be extracted from $|\gamma\rangle \otimes |0\rangle$ for $\gamma' < \gamma$ with nonzero probability: one has to sacrifice arbitrarily small amplitude of the input coherent state in order to extract the end state.

In[81]:=
```
BesselIasymp[k_, a_] := (a/2)^k / k!;
Ckasymp = BesselIasymp[k, a] Exp[-a];
Pkasymp = DiscreteConvolve[Ckasymp, convexpr, k, l] /. {a → (1 - ϵ) a, l → k};
Print["Limit of Pk/Ck for k→∞: ",
 Limit[Pkasymp/Ckasymp // Expand, k → ∞, Assumptions → {ϵ < 1, ϵ > 0}]]
```

```
Limit of Pk/Ck for k→∞: 0
```

[1] Francesco Mainardi, Fractional Calculus and Waves in Linear Viscoelasticity, Francesco Mainardi